
Abstract

NCREGRID is a tool for data transformation of gridded 2- and 3-dimen-
sional (spatial) geophysical / chemical scalar fields between grids of different
resolutions. The program handles data on rectangular latitude / longitude
grids (not necessarily evenly spaced and / or monotone) and vertical pres-
sure hybrid grids of arbitrary resolution. The input / output data format is
netCDF. The NCREGRID core is a recursive algorithm (NREGRID) which
is applicable to arbitrary (curvilinear) grids (with orthogonal axes) of any
dimension and not restricted to geophysical / chemical global hybrid grids.
NCREGRID is available under the GNU public license.

This manual is part of the electronic supplement of the article
“Technical Note: Recursive rediscretisation of geo-scientific data in
the Modular Earth Submodel System (MESSy)” in Atmos. Chem.
Phys. (2006), available at: http://www.atmos-chem-phys.org

∗Max Planck Institute for Chemistry, P.O.Box 3060, 55020 Mainz, Germany

1 Introduction

In 3-dimensional global geophysical / geochemical modelling the solving of sets of
differential equations describing geophysical / chemical properties in the space time
domain is often performed by discretisation with the Eulerian approach, i.e., on a
discrete grid of spatial coordinates. Model input parameters need to be available
on 2- or 3-dimensional (2D, 3D) grids of various resolutions. The horizontal grid
space usually comprises geographical latitude and longitude. Especially in 3D global
atmospheric models the vertical pressure (p) coordinate is often defined by hybrid
levels (with index i) of the form

p(i, x, y, t) = ha(i) · p0 + hb(i) · ps(x, y, t) , (1)

where ps is the surface pressure, p0 is a constant reference pressure, and ha and hb are
the dimensionless hybrid coefficients. This representation in a curvilinear coordinate
system (dependent on longitude x, latitude y, and time t) allows a terrain following
vertical coordinate, if ha = 0 and hb = 1 for the lowest level (surface level).

A wide variety of grid resolutions is used in different models and for different
purposes. Moreover, global geophysical / chemical data (including observations) are
provided in different spatial resolutions. As a consequence, data sets have often to be
adapted to a specific grid resolution. For this purpose, NCREGRID was developed.

NCREGRID can be used as a stand-alone program, and / or coupled as an
import interface to a model, to regrid automatically the input from an arbitrary
grid space onto the required grid resolution.

The regridding is performed by a recursive algorithm, which is applicable to
arbitrary orthogonal (also curvilinear) grids of any dimension, and not restricted to
geo-hybrid-grid structures as described above.

2 License

NCREGRID is free software; it can be redistributed and / or modified under the
terms of the GNU General Public License as published by the Free Software Foun-
dation; either version 2 of the License, or any later version, and under additional
agreements for scientific software as described in the file LICENSE.txt delivered with
this distribution.

This program is distributed in the hope that it will be useful, but WITHOUT
ANY WARRANTY; without even the implied warranty of MERCHANTABILI-
TY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU General Public
License for more details.

A copy of the GNU General Public License (GPL.txt) should have been shipped
along with this distribution; if not, it can be received from the Free Software Foun-
dation, Inc., 59 Temple Place - Suite 330, Boston, MA 02111-1307, USA.

3 Prerequisites

NCREGRID is written in Fortran95, thus a Fortran95 compiler is required for instal-
lation of the software. (A Fortran90 compiler which is capable to handle initialisation
statements in declaration lines should do as well.) Furthermore, (g)make is used for
building the software.

Since the input and output format is netCDF (http://www.unidata.ucar.edu/
packages/netcdf/), the Fortran90 version (i.e., at least netCDF - Version 3.6.0) of
the netCDF library is required.

NCREGRID comprises the f2kcli command line interface (http://www.winteracter.
com/f2kcli/index.htm) for the stand-alone version.

4 Installation

The installation is straightforward:

1. unpack the archive:
uncompress ncregrid.tar.Z

tar xvf ncregrid.tar

2. change to the subdirectory ./ncregrid:
cd ncregrid

3. configure NCREGRID according to your system:
./configure VARIABLE=VALUE . . .
with possible VARIABLEs:
F90 name of the Fortran90/95 compiler (optional)
F90FLAGS Fortran90/95 compiler options (required)

(e.g., option for invoking the cpp preprocessor)
NC INC (absolute) path where netcdf.mod

is located (required)
NC LIB (absolute) path where libnetcdf.a

is located (required)
Platform / Compiler specific notes can be found in the README file of the
distribution.

4. build the executable and the modules:
gmake

5. move the executable and the modules to the ./bin and ./include subdirec-
tories, respectively:
gmake install

6. clean-up the source directory:
gmake clean

The executable ncregrid should now be available in the ./bin subdirectory, the
modules in the ./include subdirectory. The status after unpacking the archive can
be achieved with gmake distclean.

5 Tested Systems

NCREGRID has been tested so far on / with the following platforms / compilers:

• PC (i686) Linux-2.4.18 (SuSE 8.0 - 10.0) with
lf95 (Lahey/Fujitsu Fortran 95 Express Release L6.10a)

• Dec-Alpha OSF1-4.0f with
f90 (Compaq Fortran 90 (formerly DIGITAL Fortran 90))

• Dec-Alpha OSF1-5.1a with
f90 (Compaq Fortran 90 (formerly DIGITAL Fortran 90))

• Silicon Graphics Origin 2100

• Windows 98 with Compaq Visual Fortran

• IBM-AIX

• SGI Origin 3800, IRIX6.5 with MIPSpro 7

• NEC-SX6

• PC (i686) Linux-2.4.20 (SuSE 8.2) with
NAG-f90

• Apple G5, OS v10.4.4 with xlf90 v8.1

If you succeed to build NCREGRID on platforms which are not listed here, please
send an e-mail with system / compiler information, potential changes / extensions
to the source code, and / or the Makefile to ncregrid@p-joeckel.de.

6 Usage

NCREGRID can be applied in two different modes:

• Stand-alone mode for the rediscretisation of data files in netCDF

• Interface mode (coupled to a model) for automatic rediscretisation during data
import from netCDF files

In both modes, NCREGRID is basically applied in 4 steps:

1. Analysis of the netCDF file containing the data which should be regridded
(infile). This can for instance be done with:
ncdump -h infile
Required are the names of the netCDF variables spanning the grid (such as
latitude, longitude, surface pressure, reference pressure, hybrid-coefficients,
time), and the names of the variables which should be regridded. Note that
this step is left to the user, in order to be independent of specific netCDF
conventions. Note further that netCDF and NCREGRID are case-sensitive.

2. Analysis of the output grid structure:

• The output grid structure is available in a netCDF file:
The information from that file (grdfile) can be extracted from this file in
analogy to step 1. above.

• NCREGRID is used in interface mode:
The output grid information is provided by the grid-specification interface
routine (see section 9.1)

• The output grid structure is not available:
The output grid information can be written with an appropriate text
editor in the CDL-syntax (see netCDF-manual). From this, a netCDF
file can be generated by
ncgen -b -o grdfile cdl-file
and used as grdfile, as above.

3. Summary of all required information in a namelist:
The NCREGRID-namelist structure is described in detail in section 8.

4. Start NCREGRID:

• NCREGRID in stand-alone mode:
ncregrid namelist-file

• NCREGRID in interface mode:
start the executable with NCREGRID linked
Detailed information about the usage of NCREGRID from Fortran90 /
Fortran95 code is provided in section 9.

7 Regridding Types

The core of NCREGRID, NREGRID (see section 10.4), is designed to rediscretise
arbitrary distributions from one grid to another, without adding information (e.g.,
if regridding from a coarser to a finer grid), or reducing more information than is
lost anyway (e.g., if regridding from a finer to a coarser grid). Thus, no inter- /
extrapolation methods are applied! The data fields are purely redistributed,

assuming constant values within the grid box, represented by the grid box mid point.
For this, intensive and extensive variables are distinguished.

Accordingly, the following regridding types (RG TYPE) are supported at present
for the regridding of variables (scalar fields!) between geo-hybrid grids:

• INT is suitable for intensive quantities, such as tracer mass mixing ratios, or
temperature fields. This option conserves the global weighted average of a
scalar field during the regridding procedure.

• EXT is suitable for extensive quantities, such as tracer masses, or tracer emis-
sion maps (2D). This option conserves the global unweighted sum over all grid
boxes of a scalar field.

• IDX is suitable for regridding index distributions, i.e., scalar fields with discrete
values, where averaging is not defined. The index-regridding returns for a given
destination grid-box the index, which has in all overlapping source grid-boxes
the largest relative contribution.

• IFX is similar to IDX, however it returns a variable extended by one dimension,
whereby the additional dimension is along the index range. A given slice along
the new index-dimension contains the fraction of that index in the respective
grid-box.

8 Namelist Control

The regridding procedure of NCREGRID is controlled by a namelist (and some
optional control parameters specified at the subroutine call of REGRID CONTROL, if
NCREGRID is used in interface mode). The syntax of a NCREGRID-namelist is:

&REGRID

variable = value,

...

/

The dots indicate a list of further namelist entries. Several namelists can be con-
catenated into one namelist-file. These namelists are then processed by NCREGRID
in sequence. Table 1 gives an overview of the possible namelist entries with their
meanings.

8.1 Grid specification

In order to allow netCDF-files to be as general as possible, and not to be restricted to
specific netCDF-conventions, the input grid (read from the netCDF file infile) has to
be specified by the user via the namelist. With the namelist variables i lat(i/m),
i lon(i/m), i hya(i/m), i hyb(i/m), i ps, and i p0 a 3D input grid (spatial) is
fully described. The output grid (g ...) is specified in analogy, and read from the
grdfile. The output is written to the netCDF file outfile.

INPUT OUTPUT Description

infile outfile input/output netCDF file
grdfile netCDF file with output grid
i lat(m/i) g lat(m/i) name of latitude axis
i latr g latr range of latitude axis
i lon(m/i) g lon(m/i) name of longitude axis
i lonr g lonr range of longitude axis
i hya(m/i) g hya(m/i) name of hybrid-A-coeff. (ha)
i hyar g hyar range of hybrid-A-coeff.
i hyb(m/i) g hyb(m/i) name of hybrid-B-coeff. (hb)
i hybr g hybr range of hybrid-B-coeff.
i time(m/i) g time(m/i) name of time axis
i ps g ps name/value of surface pressure
i p0 g p0 name/value of reference pressure
i t o t input/output time step control
g t grid time step control
var variable list
pressure vert. regridding in pressure coord.
input time output file gets input time axis

Table 1: List of NCREGRID namelist variables. With (...m/...i) the box mid
point coordinates (...m) and/or the box interfaces (...i) are specified, respectively.
i t(4), o t(3), and g t(3) are integer vectors, pressure and input time are of
type logical, i/g latr, i/g lonr, i/g hyar, and i/g hybr are real vectors of length
2, all others are character/string variables.

For the regridding procedure, the interfaces (i...) of the grid boxes are required
(except for i/g timei). If the respective data is not available in the infile / grdfile,
or the entries are not present in the namelist, the interface values are internally
calculated from the corresponding mid-box values, assuming that the interfaces are
half-way between the mid-points. The outer interfaces are calculated using the
same distance between outermost mid-point and corresponding inner interface. If
this calculation of the outer interfaces is not applicable, the user can specify them
with the namelist variables i latr, i lonr, i hyar, i hybr for the input grid, and
with g latr, g lonr, g hyar, g hybr for the output grid, respectively. For example,

...

i_latm = ’latitude’,

i_latr = -90.0, 90.0,

...

in the namelist ensures that the outer input grid latitude interfaces (calculated from
the mid-box latitudes with name latitude) are at −90.0◦ and 90.0◦. Note that in
case of i/g hyar the order of parameters is relevant, since the hybrid-A-coefficients
(see Eq. (1)) are not monotone. Therefore, in

i_hyar = a1, a2,

a1 refers to the highest grid level (corresponding to the smallest hybrid-B-coefficient
(!)), and a2 to the lowest grid level (corresponding to the largest hybrid-B-coefficient
(!)), respectively.

Since the core of NCREGRID (NREGRID, see section 10.4) performs regridding
in rectangular grid space without weighting, the latitude axis intervals have to be
weighted according to the surface. Therefore, NCREGRID uses internally

sin(latitude) (2)

for regridding along the latitude axes (see messy ncregrid geohyb.f90, SUBROUTINE
GEOHYBGRID AXES).

If the interface variables are specified in the namelist, but not the corresponding
mid-points, the latter are internally calculated, assuming that the mid-points are
half-way between the interfaces. The mid-points, however, are not used by the
regridding procedure.

Dimensions defined for the input grid (infile), but omitted for the output grid
(grdfile) are treated as invariant (see section 10.2).

8.2 Syntax of the namelist variable var

With the namelist variable var the user specifies the scalar fields of the infile (which
have to be on the specified input-grid) that should be regridded. For output to the
outfile, the variables can be renamed and scaled. Moreover, the regridding type (see
section 7) can be assigned. The syntax is

var = ’[new_name=]name[:RG_TYPE][,scale]; ...’,

whereby the dots indicate a list of further variables. name is the variable name in
infile, new name is the variable name in outfile, RG TYPE is the regridding type
(see section 7), and scale is the scaling factor. The order of :RG TYPE and ,scale
is arbitrary.

If new name is omitted, the variable is not renamed. If scale is omitted, the
variable is not scaled. If RG TYPE is omitted, NCREGRID checks the infile for the
variable attribute RG TYPE. If this attribute is not set, or the value is not recognised,
NCREGRID takes INT as default (see section 7).

If the namelist variable var is not specified at all, NCREGRID scans the infile for
all variables on the specified input grid. Renaming and scaling are not performed.
The regridding type is set to INT, unless the variable attribute RG TYPE in infile is
specified.

8.3 Time control

With the time control namelist variables i t, g t and o t the user specifies the
time-steps of netCDF-variables for regridding. The syntax is

i_t = itmin,itstep,itmax,itret, ! default: 1, 1, 0, 0

g_t = gtmin,gtstep,gtreset, ! default: 1, 1, 0

o_t = otstart,otstep,otdummy ! default: 1, 1, 0

where itmin, itstep, itmax, itret, gtmin, gtstep, gtreset, otstart, and otstep are integers
The default settings are listed above. The third entry of o t (otdummy) is currently
not used.

NCREGRID resets automatically itstep, gtstep, and / or otstep to 1, if 0 is
specified in the namelist; itmax and itret are set to itmin, if not specified in the
namelist. The overall consistency of all time step control parameters is checked
by NCREGRID, and documented by the output of error / warning messages, if
required.

The input variables are regridded between itmin and itmax with a step size of
itstep. The regridded data of time step itret are returned to the REGRID CONTROL

subroutine call (see section 9.2). Thus, itret has no meaning in the stand-alone mode
of NCREGRID.

For the destination grid of the variables at the input time steps (itmin, it-
min+itstep, itmin+2×itstep, . . . , itmax) the grid from grdfile is used at time step
gtmin, gtmin+gtstep, gtmin+2×gtstep, . . . , respectively. If gtreset is reached, the
grdfile time step is reset to gtstart again, etc. (This allows for instance the regridding
of 60 time steps of monthly averaged data (in infile), to a grid (in grdfile) which is
only known climatologically, i.e., containing 12 monthly averages.)

And finally, the regridded data is written to the output file with time steps
otstart, otstart+otstep, otstart+2×otstep, . . . . This is needed, e.g., if itstep is not 1,
but the outfile should contain a continuous time series.

With the namelist variable input time the time axis specification in the output
file (outfile) is set. Per default (input time = T) the outfile time axis is the same as
in infile. Otherwise, (input time = F) the grdfile time axis (or interface time axis
in interface mode) is taken for outfile. Additionally, in interface mode the output
time stepping (o t) is set to the infile time stepping (in case of input time=T), or
to the interface time stepping (in case of input time = F), respectively.

8.4 Vertical axis specifications

NCREGRID is capable to handle all cases of vertical pressure axes, such as

• hybrid pressure axes (ha 6= 0, hb 6= 0);

• constant pressure axes (hb = 0); i/g hybi/m omitted in namelist

• sigma levels (ha = 0); i/g hyai/m omitted in namelist

8.4.1 Surface and reference pressure

If the surface pressure and/or reference pressure is / are not contained in infile and /
or grdfile, respectively, or they should not be used, it is possible to specify a constant
value, for example:

i_p0 = ’101325.0 Pa’

The syntax is the same for g p0, i ps, and g ps. Note that in these cases the
unit must be chosen such that surface pressure, reference pressure and the hybrid-
coefficients are consistent (see Eq. (1)). The unit (’Pa’ in the specification above) is
only converted to the netCDF variable attribute units in the output file, but not
used internally for automatic unit conversions!

8.4.2 Vertical regridding coordinates

Calculation of the vertical overlap of grid-boxes between infile and grdfile is internally
performed in sigma-coordinates per default

σ(i) = p(i, x, y, t)/ps(x, y, t) , (3)

in order to avoid conservation problems in case the source and destination surface
pressure fields are different. However, a vertical regridding in pressure coordinates
can be enforced, if the variable pressure = T is specified in the namelist. Note,
however, that in such cases input and output pressure levels must have the same
units!

8.5 Namelist examples

Examples of NCREGRID namelists can be found in the
./ncregrid/examples/namelist subdirectory of this distribution.

9 Interface Mode

The usage of NCREGRID in interface mode, i.e., linked to another program, requires
two steps:

• Specification of the destination grid in the Fortran95 code, and

• calling the regridding procedure from the Fortran95 code. This can be achieved
by using the level 1 and level 2 interface routines described below.

9.1 Destination grid specification

For NCREGRID in interface mode (e.g., as part of a model) a specification of the
destination grid in the Fortran95 code is required, which provides the information as
alternative to the specification via the grdfile entry in the namelist (see section 8).
This definition section is contained in the module messy ncregrid interface.f90.
An example can be found in the ./ncregrid/examples/interface subdirectory.
The grid definition module provides two subroutines:

• SUBROUTINE INTERFACE GEOHYBGRID: returns the grid information as a For-
tran95 structure (TYPE geohybgrid, see messy ncregrid geohyb.f90, sec-
tion 10.3)

• SUBROUTINE INTERFACE SET MESSAGEMODE: controls the degree of diagnostic
output of NCREGRID; The following parameters can be used, i.e., summed
(MSGMODE = ... + ... + ...):
MSGMODE S silent
MSGMODE E error messages
MSGMODE VL little verbose
MSGMODE W warning messages
MSGMODE VM medium verbose
MSGMODE I info messages

9.2 Level 1 interface for calling NCREGRID

The level 1 interface routine for calling NCREGRID from Fortran95 code provides
access to all controllable features: SUBROUTINE REGRID CONTROL. The following ex-
ample shows a typical application:

!------------------------------------------------------------------------
SUBROUTINE MY_SUBROUTINE(...)

! NCREGRID INTERFACE
USE messy_ncregrid_control, ONLY: RG_CTRL, RG_NML, RG_STATUS &

, RG_SCAN, RG_PROC, RG_STOP &
, NML_NEXT,NML_STAY &
, RGSTAT_START, RGSTAT_CONT &
, RGSTAT_STOP &
, REGRID_CONTROL

USE messy_ncregrid_nc, ONLY: ncvar
USE messy_ncregrid_geohyb, ONLY: geohybgrid

! (OPTIONAL) OUTPUT OF REGRIDDING PROCEDURE
TYPE (ncvar), DIMENSION(:), POINTER :: var ! list of variables
TYPE (geohybgrid) :: grid ! output grid info

! MY VARIABLES
CHARACTER(LEN=100) :: filename ! name of namelist file
! ...

! MY CODE

! NAME OF FILE WITH NAMELIST(S)
filename = ’test.nml’

! START REGRIDDING
! EXAMPLE: REGRIDDING ALL VARIABLES IN ALL NAMELISTS
RG_CTRL = RG_PROC ! IMMEDIATE REGRIDDING
RG_NML = NML_NEXT ! READ NEXT NAMELIST FROM FILE
!
DO ! endless DO loop (must be terminated with EXIT)

CALL REGRID_CONTROL(RG_CTRL, RG_NML, RG_STATUS &
,TRIM(filename) &
!,iounit = 10 & ! DEFAULT = 17
!,lrgx=..., lrgy=..., lrgz=... & ! DEFAULT:
!,tmin=..., tmax=... & ! namelist
!,tstep=..., tret=... & ! entry
,var = var &
,grid = grid &
)

IF (RG_STATUS == RGSTAT_STOP) THEN
WRITE(*,*) ’END OF NAMELIST FILE REACHED !’
EXIT

END IF

! USE THE REGRIDDED VARIABLES (var) AND THE OUTPUT GRID (grid) HERE

END DO
! END REGRIDDING

END SUBROUTINE MY_SUBROUTINE
!------------------------------------------------------------------------

In the second example, regridding is performed only on special conditions, whereby
these conditions depend on the variables or the grid of the infile. This can be used,
e.g., if only a special subset of variables should be regridded, whereby the subset
depends on the model state.

!------------------------------------------------------------------------
SUBROUTINE MY_SUBROUTINE(...)

! NCREGRID INTERFACE
USE messy_ncregrid_control, ONLY: RG_CTRL, RG_NML, RG_STATUS &

, RG_SCAN, RG_PROC, RG_STOP &
, NML_NEXT,NML_STAY &
, RGSTAT_START, RGSTAT_CONT &
, RGSTAT_STOP &
, REGRID_CONTROL

USE messy_ncregrid_nc, ONLY: ncvar
USE messy_ncregrid_geohyb, ONLY: geohybgrid

! (OPTIONAL) OUTPUT OF REGRIDDING PROCEDURE
TYPE (ncvar), DIMENSION(:), POINTER :: var ! list of variables
TYPE (geohybgrid) :: grid ! output grid info

! MY VARIABLES
CHARACTER(LEN=100) :: filename ! name of namelist file
! ...

! MY CODE

! NAME OF FILE WITH NAMELIST(S)
filename = ’test.nml’

! START REGRIDDING
! EXAMPLE: REGRIDDING CONTROLLED BY INPUT VARIABELS OR
! INPUT-GRID (BOTH SPECIFIED IN NAMELIST)
RG_CTRL = RG_SCAN ! SCAN-MODE, NO REGRIDDING
RG_NML = NML_NEXT ! READ NEXT NAMELIST FROM FILE
!
DO ! endless DO loop (must be terminated with EXIT)

CALL REGRID_CONTROL(RG_CTRL, RG_NML, RG_STATUS &
,TRIM(filename) &
!,iounit = 10 & ! DEFAULT = 17
!,lrgx=..., lrgy=..., lrgz=... & ! DEFAULT:
!,tmin=..., tmax=... & ! namelist
!,tstep=..., tret=... & ! entry
,var = var &
,grid = grid &
)

IF (RG_STATUS == RGSTAT_STOP) THEN
WRITE(*,*) ’END OF NAMELIST FILE REACHED !’
EXIT

END IF

IF (CONDITION_ON_VAR(var).OR.CONDITION_ON_GRID(grid)) THEN
! REGRID
WRITE(*,*) ’CONDITION IS TRUE FOR VARIABLE OR GRID !’

IF (RG_CTRL == RG_SCAN) THEN
WRITE(*,*) ’REGRIDDING ACCORDING TO CURRENT NAMELIST ...’
RG_NML = NML_STAY ! DO NOT READ NEXT NAMELIST FROM FILE
RG_CTRL = RG_PROC ! PERFORM REGRIDDING

ELSE
WRITE(*,*) ’REGRIDDING DONE !’
!!
!! USE THE REGRIDDED VARIABLES AND THE OUTPUT GRID HERE
!! POSSIBLY CLEAN ’grid’ and ’var’
!! POSSIBLY EXIT ENDLESS DO LOOP HERE LIKE THIS:
!RG_CTRL = RG_STOP
!! this will terminate NCREGRID
!! (close files, clean memory)
!! at the next call within the DO-loop
!!

WRITE(*,*) ’SCANNING NEXT NAMELIST ...’
RG_NML = NML_NEXT
RG_CTRL = RG_SCAN

END IF
ELSE

WRITE(*,*) ’CONDITION IS FALSE FOR VARIABLE OR GRID !’
WRITE(*,*) ’TRYING NEXT NAMELIST ...’

END IF

END DO
! END REGRIDDING

END SUBROUTINE MY_SUBROUTINE
!------------------------------------------------------------------------

The variable filename specifies the file containing the namelist(s) to be read.
The default I/O unit for the namelist file is 17. This can be overwritten with the op-
tional integer parameter iounit. The parameters RG CTRL, RG NML, and RG STATUS

are predefined integer parameters, used to control the regridding procedure:

• RG CTRL switches the regridding procedure. Set to

– RG SCAN, no regridding is performed; the variable(s) and the grid are read
from the infile as specified in the namelist (see section 8) and returned in
var and grid, respectively.

– RG PROC, the regridding procedure is switched on, and the resulting vari-
able(s) and grid are returned in var and grid, respectively.

– RG STOP, NCREGRID is terminated (closing files, cleaning memory).

• RG NML controls the namelist file handling. Set to

– NML NEXT, the next namelist is read from the namelist file before scanning
/ regridding.

– NML STAY, the current namelist is used again for scanning / regridding.

• RG STATUS is returned by the subroutine. It is

– RGSTAT START, if the namelist file (filename) was just opened and the
first namelist read.

– RGSTAT CONT, as long as the end of the namelist file is not reached.

– RGSTAT STOP, if the end of the namelist file has been reached, or RG CTRL

has been set to RG STOP.

With the optional logical parameters lrgx, lrgy, and lrgz set to .FALSE.,
regridding in longitudinal, latitudinal, and vertical direction, respectively, can be
switched off, separately. The default for all three parameters is .TRUE..

NCREGRID internally loops over the time-steps of the input netCDF file (infile)
as specified by the namelist entry i t (see section 8). This can be overwritten by
the optional integer parameters tmin, tmax, tstep, and tret. The loop is then
from tmin to tmax with a step-size of tstep. The grid and variable(s) at position
tret are returned in grid and var, respectively. Table 2 gives a complete list of the
parameters of the SUBROUTINE REGRID CONTROL parameters.

9.3 Level 2 interface for calling NCREGRID

To facilitate the use of NCREGRID in interface mode, the module
messy ncregrid tools.f90 in the subdirectory ./examples/tools provides some
additional high level (level 2) interface routines.

9.3.1 SUBROUTINE RGTOOL READ NCVAR

This subroutine reads / regrids a single netCDF-variable (TYPE ncvar, see sec-
tion 10.3) according to a given namelist. The parameters are

SUBROUTINE RGTOOL_READ_NCVAR(iou, nmlfile, vname, t, var &
,grid, lrg, lrgx, lrgy, lrgz, lok)

INTEGER, INTENT(IN) :: iou ! logical I/O unit
CHARACTER(LEN=*), INTENT(IN) :: nmlfile ! namelist file
CHARACTER(LEN=*), INTENT(IN) :: vname ! variable name
INTEGER, INTENT(IN) :: t ! netCDF time step
TYPE(ncvar) :: var ! nc-variable
TYPE (geohybgrid), OPTIONAL :: grid ! output grid info
LOGICAL, INTENT(IN), OPTIONAL :: lrg ! regrid really ?
LOGICAL, INTENT(IN), OPTIONAL :: lrgx ! regrid in x
LOGICAL, INTENT(IN), OPTIONAL :: lrgy ! regrid in y
LOGICAL, INTENT(IN), OPTIONAL :: lrgz ! regrid in z
LOGICAL, INTENT(OUT), OPTIONAL :: lok ! OK?

• iou is the I/O unit for opening the namelist file.

• nmlfile is the name of the file which contains the regridding namelist.

• vname is the name of the variable (cf. section 8.2).

• t is the step along the time dimension of vname.

• var is the netCDF-variable output.

• grid is the optional output of the grid-structure of var.

• lrg is an optional switch for the regridding process, which is .TRUE. (default)
for regridding, and .FALSE. for import of the data on the input grid.

• lrgx is an optional switch for regridding along the longitudinal dimension
(default: .TRUE.).

• lrgy is an optional switch for regridding along the latitudinal dimension (de-
fault: .TRUE.).

• lrgz is an optional switch for regridding along the vertical dimension (default:
.TRUE.).

• lok is an optional status flag which is .TRUE. on success, and .FALSE. other-
wise (e.g., if the namelist file was not found, or if the variable was not in the
infile).

9.3.2 SUBROUTINE RGTOOL READ NCFILE

This subroutine reads / regrids a complete netCDF-file according to a given namelist.
The parameters are

SUBROUTINE RGTOOL_READ_NCFILE(iou, nmlfile, fname, t, var &
,grid, lrg, lrgx, lrgy, lrgz, lok)

INTEGER, INTENT(IN) :: iou ! logical I/O unit
CHARACTER(LEN=*), INTENT(IN) :: nmlfile ! namelist file
CHARACTER(LEN=*), INTENT(IN) :: fname ! filename
INTEGER, INTENT(IN) :: t ! netCDF time step
TYPE(ncvar), DIMENSION(:), POINTER :: var ! nc-variables
TYPE (geohybgrid), OPTIONAL :: grid ! output grid info
LOGICAL, INTENT(IN), OPTIONAL :: lrg ! regrid really ?
LOGICAL, INTENT(IN), OPTIONAL :: lrgx ! regrid in x
LOGICAL, INTENT(IN), OPTIONAL :: lrgy ! regrid in y
LOGICAL, INTENT(IN), OPTIONAL :: lrgz ! regrid in z
LOGICAL, INTENT(OUT), OPTIONAL :: lok ! OK?

• iou is the I/O unit for opening the namelist file

• nmlfile is the name of the file which contains the regridding namelist.

• fname is the name of the input netCDF file (infile, cf. section 8).

• t is the step along the time dimension specified in the namelist.

• var is the list of netCDF output variables list.

• grid is the optional output of the grid-structure of the variables in var.

• lrg is an optional switch for the regridding process which is .TRUE. (default)
for regridding, and .FALSE. for import of the data on the input grid.

• lrgx is an optional switch for regridding along the longitudinal dimension
(default: .TRUE.).

• lrgy is an optional switch for regridding along the latitudinal dimension (de-
fault: .TRUE.).

• lrgz is an optional switch for regridding along the vertical dimension (default:
.TRUE.).

• lok is an optional status flag which is .TRUE. on success, and .FALSE. other-
wise (e.g., if the namelist file was not found, or if the input netCDF file was
not found).

9.3.3 SUBROUTINE RGTOOL CONVERT

This subroutine is used to convert the data of a netCDF-variable (TYPE ncvar, see
section 10.3) into a 4D array of type REAL. The parameters are

SUBROUTINE RGTOOL_CONVERT(var, dat, grid, order)

TYPE (ncvar), INTENT(IN) :: var ! nc-variable
REAL, DIMENSION(:,:,:,:), POINTER :: dat ! data array
TYPE (geohybgrid), INTENT(IN) :: grid ! grid information
CHARACTER(LEN=4), INTENT(IN), OPTIONAL :: order ! axis order in data

• var is the netCDF-variable to be converted.

• dat is the 4D array (NULL-pointer !) which carries the result.

• grid is the underlying grid-structure of var.

• order is an optional 4-character string, describing the axis order of dat. The
default is ’xyzn’, where x denotes longitude, y latitude, and z the vertical di-
rection. n is the parameter-axis, e.g., the index range for IXF-type regridding.

9.3.4 SUBROUTINE RGTOOL G2C

This subroutine is used to convert the spatial coordinates of a grid-structure (TYPE
geohybgrid, see section 10.3) into 1D- and 2D-arrays. The parameters are:

SUBROUTINE RGTOOL_G2C(grid, hyam, hybm, p0, ps &
,hyai, hybi &
,latm, lonm &
,lati, loni, nlat, nlon, nlev)

TYPE (geohybgrid), INTENT(IN) :: grid
REAL, DIMENSION(:), POINTER, OPTIONAL :: hyam ! hybrid-A-coeff.
REAL, DIMENSION(:), POINTER, OPTIONAL :: hybm ! hybrid-B-coeff.
REAL, DIMENSION(:), POINTER, OPTIONAL :: p0 ! reference pressure
REAL, DIMENSION(:,:), POINTER, OPTIONAL :: ps ! surface pressure
REAL, DIMENSION(:), POINTER, OPTIONAL :: hyai ! hybrid-A-coeff.
REAL, DIMENSION(:), POINTER, OPTIONAL :: hybi ! hybrid-B-coeff.
REAL, DIMENSION(:), POINTER, OPTIONAL :: latm ! latitude

REAL, DIMENSION(:), POINTER, OPTIONAL :: lonm ! longitude
REAL, DIMENSION(:), POINTER, OPTIONAL :: lati ! latitude
REAL, DIMENSION(:), POINTER, OPTIONAL :: loni ! longitude
INTEGER, INTENT(IN), OPTIONAL :: nlat
INTEGER, INTENT(IN), OPTIONAL :: nlon
INTEGER, INTENT(IN), OPTIONAL :: nlev

• grid is the grid structure.

• hyam is the list of hybrid-A-coefficients (box-mid).

• hybm is the list of hybrid-B-coefficients (box-mid).

• p0 is the reference pressure (1D-array of length 1).

• ps is the surface pressure field (2D).

• hyai is the list of hybrid-A-coefficients (interfaces).

• hybi is the list of hybrid-B-coefficients (interfaces).

• latm is the list of latitudes (box-mid).

• lonm is the list of longitudes (box-mid).

• lati is the list of latitudes (interfaces).

• loni is the list of longitudes (interfaces).

• nlat is an optional parameter specifying the number of latitudes (box-mid).

• nlon is an optional parameter specifying the number of longitudes (box-mid).

• nlev is an optional parameter specifying the number of levels (box-mid).

Note: In case the grid is less than 3-dimensional (e.g., latitude × longitude), and the
INTEGER parameter/s of the missing dimension (i.e., in the example nlev) is/are
not specified, the subroutine returns a NULL-pointer for all corresponding arrays
(i.e., in the example for hyam, hybm, hyai, and hybi). If the parameter/s is/are
specified, however, the subroutine returns corresponding array/s of the requested
length (in the example nlev), filled with 0.0 (i.e., in the example hyam(1:nlev)

= 0.0, hybm(1:nlev) = 0.0, hyai(1:nlev+1) = 0.0, hybi(1:nlev+1) = 0.0).
This works accordingly for the other dimensions.

9.4 Handling counter information (time stepping)

In applications of NCREGRID in interface mode, it might be useful to trigger the
reading / regridding procedure of NCREGRID under certain conditions (events) for
reading / regridding a defined time-slice of one or more variables from a netCDF-file.
For example, one might wish to update a set of gridded boundary conditions once a
month during a multi-month model simulation. The file messy ncregrid tools.f90

contains an additional TYPE declaration, and the corresponding subroutines to
support this functionality.

9.4.1 TYPE NCRGCNT

This type provides a structure to store counter information:

TYPE NCRGCNT

CHARACTER(LEN=NCCNTMAXNLEN) :: name = ’’

INTEGER :: start = 1

INTEGER :: step = 1

INTEGER :: reset = 1

INTEGER :: current = 1

END TYPE NCRGCNT

name is used to identify the counter (e.g., in a list of counters), start contains the
minimum counter position, step the increment, and reset the maximum counter
position. The current counter position can for instance be used to read / regrid a
specific time-step of a netCDF-variable by setting the parameter t of SUBROUTINE
RGTOOL READ NCVAR or SUBROUTINE RGTOOL READ NCFILE to current.

9.4.2 SUBROUTINE RGTOOL NCRGCNT RST

This subroutine provides an interface for the automatic handling of counter infor-
mation, including

• the update of the counter, if it is triggered: The actual counter position
current is incremented by step. If the result is larger than reset, current
is reset to start. This allows cyclic counting.

• the unambiguous storage of a specific counter in a central list to be accessible
from everywhere, and to be potentially saved in an external file.

• the continuous alignment of the actual counter and its copy stored in the
central list.

The subroutine is called with the following parameters:

RGTOOL_NCRGCNT_RST(mname, start, restart, event, c, lout, linit)

CHARACTER(LEN=*), INTENT(IN) :: mname ! name of the list
LOGICAL, INTENT(IN) :: start ! .TRUE. at first time step

LOGICAL, INTENT(IN) :: restart ! .TRUE. at first time step
! of restart

LOGICAL, INTENT(IN) :: event ! .TRUE. on event
TYPE(NCRGCNT), INTENT(INOUT) :: c ! counter - struct
LOGICAL, INTENT(IN) :: lout
LOGICAL, INTENT(IN), OPTIONAL :: linit

• mname is a name identifying a sub-list of counters in the central list.

• start must be .TRUE. only at the very first step. It is used to prevent a counter
update immediately at start, if at the same time the event is triggered.

• restart must be .TRUE. only after the first initialisation of the counter list
from an external file. This is used to restore the actual counter from the list,
even if the event is not triggered.

• event is .TRUE., indicating that the counter update is triggered.

• c is the counter structure of the actual counter.

• lout is .TRUE. for diagnostic output.

• linit is an optional switch (default: .FALSE.). If set to .TRUE., it prevents
counters of the same name and mname to be updated (counter exists already
in the list), but creates those which are new.

The sequence of operations is as follows:

1. If start, restart, and event are all .FALSE., the subroutine is left immedi-
ately.

2. The algorithm locates the specified counter c by its name in the sub-list mname.
If the counter is found, the actual counter c is restored from its list entry.

3. The initialisation flag linit is checked: Only if linit is .TRUE. the presence
of the counter in the list is re-evaluated: If it exists in the list, the routine
stops with an error, since during the initialisation, the counter must not yet
exist. If the counter is new, however, it is saved in the list.

4. The algorithm checks, if the update has been triggered. Only if event is
.TRUE. and start is .FALSE., an existing actual counter and its copy in the
central list are updated, as described above. If the counter does not exist in
the list, the routine terminates with an error.

9.4.3 More on counter information

With the SUBROUTINE WRITE NCRGCNT LIST the complete central list of currently
stored counter information can be dumped into an ASCII file. Likewise, with
SUBROUTINE READ NCRGCNT LIST, the counter information can be restored from a
file previously written by SUBROUTINE WRITE NCRGCNT LIST. This functionality is
required, if counter information must be saved for later usage, as for instance, if
a model simulation is broken down into a chain simulation, whereby each chain
element must start exactly where the previous one ended.

The SUBROUTINE GET NEXT NCRGCNT can be used to loop over all currently stored
counters in the internal list, e.g., to search for a specific counter:

LOGICAL, :: last

TYPE(ncrgcnt), POINTER :: cptr

...

DO

CALL GET_NEXT_NCRGCNT(last, cptr)

IF (last) EXIT

! check cptr here

END DO

last is .TRUE., if the end of the list is reached. This indirect construction is required,
since the number of counters stored in the internal list is not known a priori.

And last, but not least, SUBROUTINE CLEAN NCRGCNT LIST is used to remove all
counter list entries, i.e., to clean up the memory.

10 Behind the Scenes of NCREGRID

10.1 Overview of the different modules

NCREGRID is hierarchically organised in a number of modules, as shown in Fig-
ure 1.

10.2 Sequence of operations

Figure 2 shows the steps between reading a namelist and the resulting output.
During the process, input and output grid are

• checked for consistency,

• reordered to achieve monotonous axes for NREGRID,

• completed (i.e., box-mids are calculated from interfaces, or vice-versa).

Furthermore, the grid-information of the input (infile) and output grid (grdfile or
INTERFACE GEOHYBGRID) is balanced, such that everything which is not specified for
the output grid is taken from the input grid. This allows, e.g., a horizontal regridding

ncregrid.f90mo_f2kcli.f90
command line
interface

READ NAMELIST
I/O CONTROL
TIME CONTROL

GRID SPECIFICATION

messy_ncregrid_interface.f90

<EMPTY>

messy_ncregrid_interface.f90

− latitude
− longitude
− ha*p0 + hb * ps

TYPE geohybgrid

messy_ncregrid_geohyb.f90

DIAGNOSTIC OUTPUT

messy_ncregrid_diag.f90

TYPE narray
TYPE axis

(recursive re−gridding)
NREGRID

messy_ncregrid_base.f90 TYPE ncdim
TYPE ncatt
TYPE ncvar
I/O of netCDF

messy_ncregrid_netcdf.f90

Fortran90
netCDF library

HIGH LEVEL
INTERFACE

messy_ncregrid_tools.f90
messy_ncregrid_control.f90

MODEL

Figure 1: Overview of the different NCREGRID modules with TYPE declarations
and dependencies. The blue shaded parts are only needed for the stand-alone mode.
The yellow shaded areas are required for the coupling to a model. Modules below
the green dashed line are independent on the special requirements of geo-hybrid
grids, and can be applied to different grid-types.

IMPORT

SWITCH

CHECK

gg

SORT

COMPLETE

gg

ggs, ggx

ggs, ggx

IMPORT

gis, gix

COMPLETE

gis, gix

SORT

CHECK

gi

BALANCE   PS

GEOHYBGRID_AXES

xivar(:)

BALANCE:_GEOHYBGRID

gis, gix ggs, ggx go EXPORTUN−SORT

BALANCE

BALANCE

BALANCE

xovar(:)

svaro(:)

pvaro(:)

%dat

qvari

IDX2FRAC

qvari

SORT

svari

PACK

pvari

%dat

xovar(:)

gi

axesCHECK dims

axesCHECK dims

UN−SORT

NAMELIST INTERFACE

IMPORT

CHECK

RENAME

SCALE

NREGRID

UN−PACK

FRAC2IDX

INFILE

SUBROUTINE CALL

VARIABLES

TIME  LOOP

INPUT GRID GRDFILE OUTPUT GRID

OUTFILE

REGRID_PS

Figure 2: Sequence of operations in NCREGRID. Boxes show operations, circles /
ellipses show information stored in structures (given are the used Fortran95 variable
names). Dashed boxes indicate operations within a loop over variables. Black
lines are used for operations on grid structures, blue lines are used for operations
on data variables, and red lines indicate connections between grid structures and
data variables. Shown in green is the core regridding step. The yellow shaded
boxes present the three possibilities for user interaction (The grid specification via
INTERFACE and the options of SUBROUTINE CALL are only accessible in the
interface mode of NCREGRID). Finally, the red boxes show the basic information
derived from the namelist.

TYPE geohybgrid
  CHARACTER(LEN=GRD_MAXSTRLEN) :: file
  INTEGER :: t
  TYPE(ncvar) :: lonm, latm, hyam, hybm, timem
  TYPE(ncvar) :: loni, lati, hyai, hybi, timei
  TYPE(ncvar) :: ps, p0
END TYPE geohybgrid

TYPE  narray
  INTEGER :: n
  INTEGER, DIMENSION(:), POINTER :: dim

END TYPE narray
  CHARACTER, DIMENSION(:), POINTER :: vc

  REAL (SP), DIMENSION(:), POINTER :: vr
  REAL (DP).DIMENSION(:),POINTER :: vd
  INTEGER (I8), DIMENSION(:), POINTER :: vi
  INTEGER (I4), DIMENSION(:),POINTER :: vb

TYPE ncdim
  CHARACTER(LEN=GRD_MAXSTRLEN) :: name
  INTEGER :: id
  INTEGER :: len
  LOGICAL :: fuid
  INTEGER :: varid
END TYPE ncdim

TYPE ncatt
  CHARACTER(LEN=GRD_MAXSTRLEN) :: name
  INTEGER :: num
  INTEGER :: xtype
  INTEGER :: varid
  INTEGER :: len

END TYPE ncatt
  TYPE (narray) :: dat

TYPE ncvar
  CHARACTER(LEN=GRD_MAXSTRLEN) :: name
  INTEGER :: id
  INTEGER :: xtype
  INTEGER :: ndims
  TYPE (ncdim), DIMENSION(:), POINTER :: dim
  INTEGER :: uid
  INTEGER :: ustep
  INTEGER :: natts
  TYPE (ncatt), DIMENSION(:), POINTER :: att
  TYPE (narray) :: dat
END TYPE ncvar

INTEGER, PARAMETER :: SP = SELECTED_REAL_KIND(6,37)
INTEGER, PARAMETER :: DP= SELECTED_REAL_KIND(13,307)
INTEGER, PARAMETER :: I4 = SLECTED_INT_KIND(4)
INTEGER, PARAMETER :: I8 = SELECTED_INT_KIND(9)

TYPE axis
  TYPE (narray) :: dat

  INTEGER :: ndp
  INTEGER, DIMENSION(:), POINTER :: dep
END TYPE axis

  LOGICAL :: lm

TYPE NCRGCNT
  CHARACTER(LEN=NCCNTMAXNLEN) :: name
  INTEGER                                                       :: start
  INTEGER                                                       :: step
  INTEGER                                                       :: reset
  INTEGER                                                       :: current
END TYPE NCRGCNT

messy_ncregrid_base.f90

messy_ncregrid_geohyb.f90

messy_ncregrid_tools.f90

messy_ncregrid_netcdf.f90

Figure 3: Type definitions (Fortran95 structures) of the different modules and their
dependencies.

of 3D (spatial) scalar fields, while conserving the vertical hybrid-coordinate system
of the source file. In a similar way, only vertical regridding can be performed.

Furthermore, if the surface pressure of the output grid is not available (or should
not be used!), the input surface pressure is pre-regridded and used for the output
grid. Similarly, the output surface pressure (in grdfile) is used, if the input surface
pressure is omitted (BALANCE GEOHYBGRID PS).

This balancing allows a flexible handling of the data for many purposes.

10.3 TYPE definitions

Figure 3 shows in detail the various TYPE definitions in the different modules and
their inter-connections:

• TYPE narray (messy ncregrid base.f90) provides a container for data fields
of arbitrary rank N (N -dimensional data) and type (float, double precision,
integer, byte, character). n is the rank N of the narray, i.e., the number

of dimensions, and dim an integer vector of length n containing the dimen-
sion length(s). The pointers point to the memory containing the field data,
organised as a 1-dimensional array. The module messy ncregrid base.f90

provides the

INTEGER FUNCTION POSITION(dim, vec)

which calculates for the element with integer position vector vec(:) the posi-
tion m in the linear data array, if the array is interpreted as an N -dimensional
array with dimension length(s) dim(:). The reverse procedure is provided by
the

SUBROUTINE ELEMENT(dim, m, vec)

This construction is required, since in Fortran95 the rank of an array cannot
be defined at runtime.

• TYPE axis (messy ncregrid base.f90) contains the discrete values of an ar-
bitrary axis. The logical lm is .TRUE. for modulo-axes. Further, ndp is the
number of dependencies, which is 1 for independent axes (such as longitude
and latitude), and > 0 for curvilinear axes (such as, e.g., pressure levels, which
depend for hybrid-grids on latitude, longitude, (and time)). The integer vector
dep(:) contains the numbers (indices) of the dependent axes, if the axes are
combined in a list of

TYPE (axis), DIMENSION(:), ALLOCATABLE / POINTER

• TYPE ncdim, TYPE ncvar, and TYPE ncatt (messy ncregrid nc.f90) pro-
vide three Fortran95 structures for handling the contents of a netCDF file
(see netCDF manual), namely dimensions, variables, and attributes, respec-
tively. This module also contains powerful, robust routines for netCDF input
/ output into / from these structures, such as

SUBROUTINE IMPORT_NCDIM(...)

SUBROUTINE EXPORT_NCDIM(...)

SUBROUTINE IMPORT_NCATT(...)

SUBROUTINE EXPORT_NCATT(...)

SUBROUTINE IMPORT_NCVAR(...)

SUBROUTINE EXPORT_NCVAR(...)

• TYPE geohybgrid (messy ncregrid geohyb.f90) defines the geo-hybrid-grid
structure, using netCDF variables (TYPE ncvar) for the grid box mid points
(...m) and the grid box interfaces (...i). t is the time-step of the current
grid, and file the netCDF filename for I/O.

• TYPE ncrgcnt (messy ncregrid tools.f90 provides a structure for the cyclic
counting, as described in section 9.4.

10.4 The recursive core of NCREGRID

The actual regridding is performed by the

RECURSIVE SUBROUTINE NREGRID(...)

in module messy ncregrid base.f90. The algorithm has to be recursive to allow
the regridding in an arbitrary number of dimensions.

On the first recursion level, some basic checks on the input are performed. NRE-
GRID compares the input / output axes and specifies a dimension as invariant, if
the output axis is undefined. For invariant dimensions, no regridding is required.
If a dimension is invariant and the respective axis is dependent (i.e., curvilinear),
the axis data is pre-regridded. The whole regridding procedure requires a special
order of the dimensions, which is calculated by NREGRID itself (on the first re-
cursion level): first, all independent, non-invariant dimensions, next all dependent,
non-invariant dimensions, and finally all invariant dimensions.

If the current dimension is non-invariant, NREGRID calculates the overlap -
matrices (as fractional overlap) between source and destination axis. For each non-
zero element of these matrices, it branches to the next recursion level (i.e., to the
next dimension).

If the current dimension is the first invariant dimension, NREGRID loops along
all remaining (invariant) dimensions and calculates the results for the output grid,
depending on the regridding type (RG TYPE, see section 7).

Since in most practical cases, regridding is performed for more than one variable
on the same grid (axes), NREGRID handles a list of variables in one step. The
innermost loop is over the number of variables, in oder to increase the performance,
i.e., to avoid recalculation of the same overlap-matrices for all variables separately.

